
Subject: Re: Cannot replicate IPV ever for India DHS 2005-06

Posted by [Liz-DHS](#) on Fri, 09 Nov 2018 19:38:38 GMT

[View Forum Message](#) <> [Reply to Message](#)

Dear User, A response from senior analyst, Dr. Kerry MacQuarrie:

Quote:

Hello,

I am reluctant to assess an IPV variable as "correct" or "incorrect" since it is really up to the researcher's discretion, depending on your analytic purpose. Are you trying to construct the spousal violence variable(s) as The DHS Program constructs them in, for example, Table 20.9 Forms of Spousal Violence (p 268) in the Cambodia DHS 2014 Final Report? If so, which numbers are you trying to match?

The code at the end of this message should work for most surveys to produce estimates of any spousal violence (emotional, physical, or sexual) in the past 12 months--the last results column of Table 20.9. The code would be a bit different to produce the first results column (EVER any spousal violence).

Here a couple of resources that I find helpful when I am trying to match DHS survey report tables.

1. The Guide to DHS Statistics, available here (see Ch. 17):

[https://www.dhsprogram.com/publications/publication-DHSG1-DH
S-Questionnaires-and-Manuals.cfm](https://www.dhsprogram.com/publications/publication-DHSG1-DHS-Questionnaires-and-Manuals.cfm)

2. DHS tutorial videos on Matching DHS Final Report Tables Part 1

(https://www.youtube.com/watch?v=tjuaPV4eC_E) and Part 2

(<https://www.youtube.com/watch?v=TovRq9TpbXU>)

In summary, the steps to matching a table are to check if you have:

1. Correct data file unit of analysis
2. Correct denominator population at risk
3. Correct variables find and understand your variables
4. Correct recoding special values
5. Correct weights
6. Correct tabulation row vs. column percent

I hope these resources help you.

Regards,

Kerry

Stata code to produce estimates of spousal violence (emotional, physical, sexual, & any of these 3) in past 12 months

```
use [CC]IR[VV]FL.DTA, clear      /*CC=country code; VV=version number*/  
gen dwt = d005/1000000
```

```
**Create composite variable "experienced ANY spousal violence in last 12 months"
```

```
/*NOTES:
```

1. Restricts sample to women selected for the domestic violence module.
 2. Restricts sample to ever-married women (currently married refers to current husband; formerly married refers to last husband)
 3. Domestic violence weight should be used for all analyses with this variable.
- */

*Emotional violence variables

```

recode d103a (1/2=1 "Yes") (0 3 =0 "No"),g(humil)
recode d103b (1/2=1 "Yes") (0 3 =0 "No"),g(threat)
recode d103c (1/2=1 "Yes") (0 3 =0 "No"),g(insult)

```

```

egen ev12=rowtotal(humil threat insult) if humil!=.
  recode ev12 (1/3=1)
  lab def yesno 0 "No" 1 "Yes"
  lab val ev12 yesno
  lab var ev12 "Experienced any emotional violence in last 12 months"

```

*Physical violence variables

```

recode d105a (1/2=1 "Yes") (0 3 =0 "No"),g(push)
recode d105b (1/2=1 "Yes") (0 3 =0 "No"),g(slap)
recode d105c (1/2=1 "Yes") (0 3 =0 "No"),g(punch)
recode d105d (1/2=1 "Yes") (0 3 =0 "No"),g(kick)
recode d105e (1/2=1 "Yes") (0 3 =0 "No"),g(burn)
recode d105f (1/2=1 "Yes") (0 3 =0 "No"),g(knife)
recode d105j (1/2=1 "Yes") (0 3 =0 "No"),g(twist)

```

```

egen pv12=rowtotal(push slap punch kick burn knife twist) if humil!=.
  recode pv12 (1/max=1)
  lab val pv12 yesno
  lab var pv12 "Experienced any physical violence in last 12 months"

```

*Sexual violence variables

```

recode d105h (1/2=1 "Yes") (0 3 =0 "No"),g(sex)
recode d105i (1/2=1 "Yes") (0 3 =0 "No"),g(sexact)
recode d105k (1/2=1 "Yes") (0 3 =0 "No"),g(psexact)

```

```

egen sv12=rowtotal(sex sexact psexact) if humil!=.
  recode sv12 (1/3=1)
  lab val sv12 yesno
  lab var sv12 "Experienced any sexual violence in last 12 months"

```

*Any spousal violence composite variable

```

egen dv12=rowtotal(ev12 pv12 sv12) if humil!=.
  recode dv12 (1/3=1)
  lab val dv12 yesno
  lab var dv12 "Experienced any spousal violence in last 12 months"

```

ta dv12

ta dv12 [iw=dwt]

Kerry LD MacQuarrie, PhD